# Maintainer: Maxime Gauduin <alucryd@archlinux.org>

pkgname=wingpanel
pkgver=3.0.3
pkgrel=2
pkgdesc='The Pantheon Top Panel'
arch=(loong64 x86_64)
url=https://github.com/elementary/wingpanel
license=(GPL3)
groups=(pantheon)
depends=(
  glib2
  gtk3
  libgala.so
  libgee
  libgranite.so
  mutter
  pantheon-notifications
)
makedepends=(
  cmake
  git
  gala
  meson
  vala
)
provides=(libwingpanel-2.0.so)
options=(debug)
_tag=f8b98a59d4d5cecd13f1c1a71f4700aeb006cc21
source=(git+https://github.com/elementary/wingpanel.git#tag=${_tag})
b2sums=(SKIP)

pkgver() {
  cd wingpanel

  git describe --tags
}

build() {
  arch-meson wingpanel build
  meson compile -C build
}

package() {
  meson install -C build --destdir "${pkgdir}"
}

# vim:set sw=2 sts=-1 et:
